INTERNATIONAL ATOMIC ENERGY AGENCY
UNITED NATIONS EDUCATIONAL, SCIENTIFIC AND CULTURAL ORGANIZATION
INTERNATIONAL CENTRE FOR THEORETICAL PHYSICS
I.C.T.P., P.O. BOX 586, 34100 TRIESTE, ITALY, CABLE: CENTRATOM TRIESTE

UNITED NATIONS INDUSTRIAL DEVELOPMENT ORGANIZATION

INTERNATIONAL CENTRE FOR SCIENCE AND HIGH TECHNOLOGY

SMR/548-18

Course on Oceanography of Semi-Enclosed Seas
15 April - 3 May 1991

"Standard Computations of Physical Seawater Properties"

F. STRAVISI
University of Trieste
Department of Theoretical Physics
Trieste, Italy

Please note: These notes are intended for internal distribution only

Università degli Studi di Trieste
Dipartimento di Fisica Teorica

Cattedra di Fisica Terrestre e Climatologia

Franco Stravisi

FORTRAN AND BASIC PROGRAMS FOR COMPUTING AND PLOTTING THE ASTRONOMIC TIDE

Rapporto Interno
FTC 89/4

Trieste 1989

FORTRAN AND BASIC PROGRAMS FOR COMPUTING AND PLOTTING THE ASTRONOMIC TIDE

Franco Stravisi
University of Trieste,
Department of Theoretical Physics

ABSTRACT. Fortran programs are given for computing the astronomic tide at Trieste. Outputs are (i) an Ascii file containing one year of hourly elevations, (ii) monthly tables with hourly elevations and (iii) monthly tables with high and low waters. A BASIC program is described for plotting a monthly graph of the astronomical tide.

The harmonic constants, seven tidal components, can be easily changed inside the program, for use in other localities. The software usually runs on a PC under MS-DOS.

1. The tidal prediction at Trieste.

The conventional harmonic method has been adopted since many years (Polli, 1949) for the computation of the high and low waters of the astronomic tide ("tidal prediction") at the Istituto Sperimentale Talassografico (IST) of Trieste. During the 60's, the use of a Doodson-Légé tide predicting machine permitted to speed-up this job. In 1976, a Fortran IV program (PREMAR) for the harmonic tidal computations, running on the CDC 6200 computer at the University of Trieste, was written at IST by Stravisi. This program has been used since that time for computing and printing tidal tables for Trieste (Stravisi, Ferraro and Luca, 1976) and other Adriatic harbours. PREMAR has been modified in 1980 in order to run on a Digital Minc-11 PC, under RT-11, at IST (Stravisi, Ferraro and Luca, 1980). In the following years, some improvements were introduced in order to compute also the nodal factors (f) and the arguments (V_o+u), (Stravisi, 1983), and for drawing monthly graphs on the IST Tektronix 4662 plotter (Stravisi, Ferraro and Luca, 1983). The 1986 revision of PREMAR (Stravisi, 1986), with some graphic improvements by Ferraro, is still in use at IST.

The recent diffusion of PC's using MS-DOS and Fortran77 required the conversion of PREMAR 1986, accounting for the different files management and taking advantage of the increased memory capabilities. The new software produces, in addition, an Ascii formatted file containing the hourly elevations of the astronomic tide. These data can be plotted on the PC screen by means of a BASIC program; a printed copy can be then obtained, avoiding the use of a plotter. In this way, tidal predictions for Trieste have been prepared first at UNEP-CIMAM, Trieste, using a PC Olivetti M24 (Stravisi, 1987, 1988), and then at the Department of Theoretical Physics of the University of Trieste, in collaboration with the Climatology Laboratory of the International Centre for Theoretical Physics, using an Olivetti M380/C (Stravisi, 1989).

The 1989 version of the software for tidal predictions is described in this report. Different features for output on printer/files have been included into a single MS-Fortran program (TIDTAB); a number of subroutines and functions are used, taken from Stravisi (1983). The BASIC program PLOMAR is used for plotting monthly graphs of the astronomic tide.

2. The harmonic method.

The astronomical tide, according to the classical harmonic model, is conventionally computed by adding a number of tidal components

$$\eta_i = f_i H_i \cos(\sigma_i t + (Vo+u)_i - g_i) \; ;$$

t is the local time, the mean amplitude H and the phase g are the "harmonic constants" computed from the tidal records and V_o+u represents the equilibrium argument at Greenwich at t=0. The angular speed σ can be considered to be constant with time. The nodal factor f, accounting for variations having a period of 18.61 years, is commonly used to modulate the tidal component amplitude, so that the model, in spite of its name, is not strictly harmonic. Tables giving numerical values of f, V_o and u for a number of tidal components and for different years can be found in the literature (Schureman, 1958). Formulas (see Tables 1,2) and codes for their computation are reported by Stravisi (1983) for the seven tidal components (M_2,S_2,N_2,K_2;K_1,O_1,P_1) that are required at Trieste, and almost everywhere, in order to represent the real tide with a good accuracy. The corresponding values for some shallow-water components of common use can be computed on the basis of the corresponding values of the principal constituent tides (Schureman, 1958).

3. Program TIDTAB and related routines.

The software for tidal predictions described in this section is normally used on PC's like Olivetti M24 and M380 with MS-DOS and Microsoft Fortran.

The main <u>PROGRAM TIDTAB</u> is listed on pages 10-23. The only inputs, from the keyboard, are the prediction year and a code (1/2/3) for different outputs. The harmonic constants for Trieste are defined in the subroutine ARCOTS; all the parameters required by the harmonic model are computed by appropriate subroutines. The printer control sequences correspond to the IBM Proprinter; 80 columns are used on UNI-A4 sheets. Run TIDTAB, enter year and code. Nodal factors and equilibrium arguments are displayed. Code = 1: the hourly elevations of the astronomical tide for the whole year are computed and stored in an Ascii formatted file; the file name is supplied from the keyboard. Data are in centimeters with one decimal digit; the reference level is zero. Code = 2: input the month number (1-12, 0 to stop). Monthly tables are printed, with the hourly values of the astronomical tide. Code = 3: input the month number (1-12, 0 to stop). The series of consecutive high and low water instants and elevations are computed by subroutine HIGLOW; monthly tables are printed. For output formats, see Stravisi (1989). Explanation comments are inserted within the program list. Headings can be changed in formats (1), 20,26,31,50. Names of the months are defined in DATA MESE; days of the week in DATA GN.

FUNCTION TIDE. Computes the astronomical tide at time T (in hours; T=0 is 0 h, 1 January), by means of seven tidal constituents. Amplitudes, angular speeds and phases are given in COMMON /FTIDE/.

FUNCTION D, DD. Compute the first and the second time derivatives of the harmonic tide, to be used for finding high and low waters.

SUBROUTINE ARCOTS. Defines the angular speeds of M_2, S_2, N_2, K_2, K_1, O_1, P_1 and the corresponding harmonic constants H,g for Trieste (Trotti, 1969). Calls for the computation of f and V_0+u for the current year, displays a summary table and puts suitable parameters in COMMON /FTIDE/. Harmonic constants for other harbours can be defined in DATA H,G.

SUBROUTINE HSPLON. Computes the mean longitudes DL of the sun, of the moon and of the lunar perigee (h, s and p in Table 1d) at a given time Y in years AC. Output values are in degrees/radians if IRD=0/1.

SUBROUTINE NLONG. Computes the longitude of the lunar node (N in Table 1d) at given time Y in years AC. Output in degrees/radians according to IRD.

SUBROUTINE INUXI. Computes five elements DX of the lunar orbit (Table 1e) at given time Y (years AC): the obliquity I of the lunar orbit with respect to earth's equator, the right ascension of the lunar intersection ν , the auxiliary terms ν' and $2\nu''$ for the tidal constituents K_1 and K_2 and the longitude ξ , in moon's orbit, of the lunar intersection. Output in degrees/radians (IRD=0/1).

SUBROUTINE VZEROU. Computes the principal portion of the argument V(t) at τ=0 (τ=180°), and the argument u at the middle of the time interval, according to formulas of Table 2. Output values are V_0+u, seven tidal components, in degrees or radians (IRD=0/1). Input times, for one year of tidal predictions, are 0 h, 1 January, given year (e.g. Y=1990.0) and the middle of the year (e.g. YH=1990.5).

Values of f and V_0+u for the years 1981-2020 are reported in Table 3; the corresponding values for S_2 are 1 and 0.

SUBROUTINE NODFAC. Computes the nodal factors f (Table 2) at the middle of the prediction period (e.g. Y=1990.5).

SUBROUTINE TIMCOF. Computes the auxiliary coefficients A, B (Table 1c) for the year Y, using constants and parameters of Table 1a,b.

SUBROUTINE HIGLOW. Computes the instants HP(I), with an accuracy better than 0.5 min, and the corresponding elevations LP(I) in centimeters, rounded to integers, of the consecutive high and low waters I=1,NHL for the given calendar year NY. Times of occurrence of the relative extremes are found by means of the Newton-Fourier iteration, using the first and the second time derivatives of the astronomic tide, applied to consecutive time intervals [t_A,t_B=t_A+S]. The method is described by Stravisi (1983) and the corresponding flowchart is reported in Fig. 1. Convergence usually occurs within three iterations.

4. BASIC program PLOMAR.

This program displays on the PC screen a monthly graph of the astronomical tide; the phases of the moon are shown at the corresponding days.

Input (i) year, (ii) name of the Ascii file created by TIDTAB (code 1), containing the hourly elevations of the astronomical tide, (iii) name of file containing the phases of the moon, and (iv) month (1-12). After the first plot, select another month by typing GOTO 290 to skip data input.

Undesired characters on the screen can be cancelled; press the <PRINT SCREEN> key to obtain a printed copy of the graph.

The PC screen is defined on line 380; SCREEN 3 defines the Olivetti high resolution (640x400 pixels). On line 1020 the cursor is cancelled; this statement can be illegal in some new versions of GW-BASIC. Lunar phases are plotted on lines 1050-1120; symbols are centered at 18 h of the corresponding day, at a mean height between tidal curves. Moon's phase data are stored in an Ascii file with the following format:

```
1990  50
01 04 11.41  2
01 11 05.58  3
01 18 22.18  4        (example of file with moon's phases)
01 26 20.21  1
02 02 19.33  2
...
```

The first row indicates the year and the corresponding total number of lunar phases. The following rows report month, day, time in hours and phase 1,2,3,4, corresponding to new moon, first quarter, full moon and last quarter respectively.

PLOMAR can be changed for use in other localities, by defining new headings, or for plotting sea level records. The vertical scale is however suitable for elevations between ± 80 cm.

REFERENCES

POLLI S. (1949): "Tabelle di previsione delle maree per il Golfo di Trieste per l'anno 1950", Ist. Sper. Talassografico Trieste, N. 249, 1-12.

SCHUREMAN P. (1958): "Manual of harmonic analysis and prediction of tides", U.S. Dept. of Commerce, Coast and Geodetic Survey, Spec. Pub. no. 98, 317 pp.

STRAVISI F. (1983): "The IT method for the harmonic tidal prediction", Boll. Oceanol. Teor. Appl., I, 3, 193-204.

STRAVISI F. (1986): "A Fortran program for the harmonic tidal prediction", Ist. Talassografico Trieste, Int. Rep. RF 14.

STRAVISI F. (1987): "Previsioni di marea per il Golfo di Trieste - 1988", FTC 87/4, 34 pp.

STRAVISI F. (1988): "Previsioni di marea per il Golfo di Trieste - 1989", FTC 88/5, 34 pp.

STRAVISI F. (1989): "Previsioni di marea per il Golfo di Trieste - 1990", FTC 89/5, 40 pp.

STRAVISI F., FERRARO S., LUCA I. (1976): "Previsioni di marea - Golfo di Trieste 1977", Ist. Sper. Talassografico Trieste, N. 527.

STRAVISI F., FERRARO S., LUCA I. (1980): "Previsioni di marea - Golfo di Trieste 1981", Ist Talassografico Trieste, N. 553.

STRAVISI F., FERRARO S., LUCA I. (1983): "Golfo di Trieste - Previsioni di marea per il 1984", Nova Thalassia, 6 suppl.

TROTTI L. (1969): "Nuove costanti armoniche per i porti adriatici e ionici", Atti Acc. Ligure Sc. Lett., 26.

) Constants

c	$= 3.844\,03 \times 10^{8}$ m	mean earth-moon distance	(1)
c_1	$= 1.495\,042\,01 \times 10^{11}$ m	mean earth-sun distance	(1)
S/E	$= 332\,488 \pm 43$	sun/earth mass ratio	(2)
M/E	$= 12\,289 \pm 4 \times 10^{-6} = 1 / 81.37$	moon/earth mass ratio	(2)
S/M	$= 2.705\,455 \times 10^{7}$	sun/moon mass ratio	
S'	$= (c/c_1)^3$ S/M $= 0.459\,875\,64$	solar factor	
e	$= 0.054\,900\,56$	eccentricity of moon's orbit	(2)
i	$= 5.145\,376\,28°$	inclination of moon's orbit to plane of ecliptic	(3)

) Time dependent parameters

n		time after 1900, in years	
e_1	$= 0.016\,751\,04 - 4.180 \times 10^{-7}\,n - 1.26 \times 10^{-11}\,n^2$	eccentricity of earth's orbit	(2)
ω	$= 23.452\,294° - 1.301\,11° \times 10^{-4}\,n$	obliquity of the ecliptic	(1)

) Time dependent auxiliary coefficients

		Numerical value, 1985	Increment, per year
A	$= S'(1 + 3/2\,e_1^2) / (1 + 3/2\,e^2)$	0.913 771 493	+ 0.000 000 900
A_1	$= \cos i \cos \omega$	0.035 676 679	− 0.000 000 187
A_2	$= \sin i \sin \omega$		
A_3	$= \cos \frac{1}{2}(\omega - i) / \cos \frac{1}{2}(\omega + i)$	1.018 819 128	− 0.000 000 108
A_4	$= \sin \frac{1}{2}(\omega - i) / \sin \frac{1}{2}(\omega + i)$	0.643 957 699	− 0.000 001 671
A_5	$= A \sin 2\omega$	0.334 316 893	− 0.000 001 429
A_6	$= A \sin^2 \omega$	0.072 478 792	− 0.000 000 761
B_1	$= [\cos \frac{\omega}{2} \cos \frac{i}{2}]^{-4}$	1.092 333 626	− 0.000 001 029
B_2	$= [A_5 + (1 - 3/2 \sin^2 i) \sin 2\omega]^{-2}$	0.897 663 509	+ 0.000 007 647
B_3	$= [A_6 + (1 - 3/2 \sin^2 i) \sin^2 \omega]^{-2}$	19.098 898 614	+ 0.000 400 363
B_4	$= [\sin \omega \cos^2 \frac{\omega}{2} \cos^4 \frac{i}{2}]^{-1}$	2.632 568 903	+ 0.000 012 547
B_5	$= 2 A_5 B_2$	0.600 208 150	+ 0.000 002 548
B_6	$= 2 A_6 B_3$	2.768 530 194	+ 0.000 028 978
B_7	$= [1 + (1 - 3/2 \sin^2 i) / A]^{-2} = B_2 A_5^2 = B_3 A_6^2$	0.100 329 862	− 0.000 000 003

) Longitude of lunar and solar elements (3)

T	time in Julian centuries (36 525 d), reckoned from Greenwich mean noon, December 31, 1899	
h	$= 279.696\,678° + 36\,000.768\,925°\,T + 3.025° \times 10^{-4}\,T^2$	mean longitude of sun
s	$= 270.437\,422° + 481\,267.892\,000°\,T + 2.525° \times 10^{-3}\,T^2 + 1.89° \times 10^{-6}\,T^3$	mean longitude of moon
p	$= 334.328\,019° + 4\,069.032\,206°\,T - 1.034\,4° \times 10^{-2}\,T^2 - 1.25° \times 10^{-5}\,T^3$	longitude of lunar perigee
N	$= 259.182\,533° - 1\,934.142\,397°\,T + 2.106° \times 10^{-3}\,T^2 + 2.22° \times 10^{-6}\,T^3$	longitude of moon's node

) Time dependent elements of the lunar orbit (3)

I	$= \arccos [A_1 - A_2 \cos N]$	obliquity of lunar orbit with respect to earth's equator
C	$= \arctan [A_3 \tan N/2]$	
ν	$= C - \arctan [A_4 \tan N/2]$	right ascension of lunar intersection
ν'	$= \arctan [(\sin 2I \sin \nu) / (A_5 + \sin 2I \cos \nu)]$	auxiliary term for K_1
$2\nu''$	$= \arctan [(\sin^2 I \sin 2\nu) / (A_6 + \sin^2 I \cos 2\nu)]$	auxiliary term for K_2
ξ	$= N + \nu - 2C$	longitude in moon's orbit of lunar intersection

1) American Ephemeris and Nautical Almanac
2) Smithsonian Physical Tables
3) Schureman (1958)

Tab. 1. Astronomical parameters of use in tidal computations (from Stravisi, 1983).

	f	V(t) (1)	u	σ (2)
M_2	$B_1 \cos^4 \frac{I}{2}$	$2\tau - 2s + 2h$	$2\xi - 2\nu$	$28.984\,104\,214 - 10.14 \times 10^{-9}\,T = 28.984\,104\,205$
S_2	1	2τ	0	30
N_2	$B_1 \cos^4 \frac{I}{2}$	$2\tau - 3s + 2h + p$	$2\xi - 2\nu$	$28.439\,729\,516 - 28.16 \times 10^{-9}\,T = 28.439\,729\,492$
K_2	$\{B_3 \sin^4 I + B_6 \sin^2 I \cos 2\nu + B_7\}^{1/2}$	$2\tau + 2h$	$-2\nu''$	$30.082\,137\,278 + 1.38 \times 10^{-9}\,T = 30.082\,137\,279$
K_1	$\{B_2 \sin^2 2I + B_5 \sin 2I \cos \nu + B_7\}^{1/2}$	$\tau + h - 90°$	$-\nu'$	$15.041\,068\,639 + 0.69 \times 10^{-9}\,T = 15.041\,068\,640$
O_1	$B_4 \sin I \cos^2 \frac{I}{2}$	$\tau - 2s + h + 90°$	$2\xi - \nu$	$13.943\,035\,575 - 10.84 \times 10^{-9}\,T = 13.943\,035\,566$
P_1	1	$\tau - h + 90°$	0	$14.958\,931\,361 - 0.69 \times 10^{-9}\,T = 14.958\,931\,360$

(1) $\tau = 15°\,t + 180°$ is the hour angle of the mean sun; t is Greenwich time in hours.
(2) Angular speed in °/h; terms in T^2, computed for T = 1, are enclosed.
Constant speeds refer to T = 0.85, year 1985.

Tab. 2. **Nodal factors, arguments and speeds of the seven principal tidal constituents (from Stravisi, 1983).**



Fig. 1. Computation of the high and low water instants, Newton-Fourier method (Subroutine HIGLOW). S=1 h, σ=S/5, ε=0.5 min (from Stravisi, 1983).

Year	M_2	N_2	K_2	K_1	O_1	P_1
1981	1.021	1.021	0.864	0.948	0.915	1.000
	117.90	287.43	185.58	2.49	119.42	349.43
1982	1.009	1.009	0.949	0.987	0.979	1.000
	218.36	299.18	183.03	1.45	221.06	349.67
1983	0.996	0.996	1.046	1.026	1.042	1.000
	319.06	311.15	182.62	1.46	321.41	349.91
1984	0.984	0.984	1.142	1.060	1.096	1.000
	60.00	323.37	184.16	2.36	60.81	350.15
1985	0.974	0.974	1.226	1.087	1.140	1.000
	136.78	298.36	189.25	4.91	134.20	349.40
1986	0.967	0.967	1.286	1.104	1.168	1.000
	238.11	310.98	193.46	6.93	232.56	349.64
1987	0.963	0.963	1.315	1.112	1.182	1.000
	339.56	323.70	198.30	9.21	330.72	349.88
1988	0.964	0.964	1.310	1.111	1.180	1.000
	81.04	336.45	203.30	11.55	68.82	350.12
1989	0.969	0.969	1.271	1.100	1.161	1.000
	158.08	311.71	209.99	14.76	141.65	349.37
1990	0.977	0.977	1.203	1.080	1.128	1.000
	259.37	324.27	213.92	16.67	240.11	349.61
1991	0.988	0.988	1.115	1.051	1.081	1.000
	0.47	336.65	216.61	18.06	339.03	349.85
1992	1.000	1.000	1.017	1.015	1.024	1.000
	101.34	348.80	217.61	18.71	78.68	350.09
1993	1.013	1.013	0.922	0.975	0.960	1.000
	177.58	323.25	218.55	19.42	154.00	349.34
1994	1.024	1.024	0.842	0.937	0.897	1.000
	277.98	334.93	215.38	18.05	256.07	349.58
1995	1.032	1.032	0.785	0.905	0.844	1.000
	18.20	346.43	210.34	15.59	359.72	349.82
1996	1.037	1.037	0.753	0.885	0.812	1.000
	118.30	357.80	204.05	12.25	104.76	350.06
1997	1.038	1.038	0.749	0.883	0.807	1.000
	193.96	331.67	199.21	9.48	185.15	349.31
1998	1.034	1.034	0.772	0.897	0.831	1.000
	294.03	343.02	192.67	5.95	290.52	349.55
1999	1.027	1.027	0.821	0.926	0.879	1.000
	34.20	354.47	187.16	3.18	34.65	349.79
2000	1.016	1.016	0.895	0.963	0.939	1.000
	134.54	6.09	183.35	1.45	137.22	350.03

Tab. 3. **Nodal factors f in the middle, and arguments**

Year	M_2	N_2	K_2	K_1	O_1	P_1
2001	1.004	1.004	0.986	1.003	1.004	1.000
	210.71	340.47	183.59	1.82	212.95	349.28
2002	0.992	0.992	1.084	1.040	1.064	1.000
	311.50	352.54	183.97	2.20	312.89	349.52
2003	0.980	0.980	1.177	1.071	1.115	1.000
	52.53	4.85	186.17	3.38	52.01	349.76
2004	0.971	0.971	1.252	1.094	1.152	1.000
	153.76	17.36	189.76	5.14	150.59	350.00
2005	0.965	0.965	1.301	1.108	1.176	1.000
	230.77	352.58	196.26	8.28	223.48	349.25
2006	0.963	0.963	1.317	1.113	1.183	1.000
	332.24	5.32	201.21	10.60	321.60	349.49
2007	0.965	0.965	1.299	1.108	1.174	1.000
	73.70	18.06	206.16	12.92	59.72	349.73
2008	0.972	0.972	1.248	1.093	1.150	1.000
	175.08	30.72	210.62	15.04	158.00	349.97
2009	0.981	0.981	1.171	1.069	1.112	1.000
	251.92	5.78	216.11	17.77	231.24	349.22
2010	0.992	0.992	1.077	1.038	1.060	1.000
	352.94	18.07	218.20	18.89	330.41	349.46
2011	1.005	1.005	0.979	1.000	0.999	1.000
	93.71	30.12	218.44	19.21	70.42	349.70
2012	1.017	1.017	0.889	0.960	0.935	1.000
	194.25	41.93	216.57	18.52	171.61	349.93
2013	1.027	1.027	0.817	0.923	0.875	1.000
	270.19	16.09	214.59	17.70	248.92	349.19
2014	1.034	1.034	0.769	0.895	0.829	1.000
	10.35	27.53	208.98	14.86	353.15	349.43
2015	1.038	1.038	0.748	0.882	0.807	1.000
	110.42	38.87	202.40	11.30	98.58	349.67
2016	1.037	1.037	0.755	0.886	0.813	1.000
	210.46	50.19	195.61	7.56	204.31	349.90
2017	1.032	1.032	0.788	0.907	0.847	1.000
	286.18	24.12	191.34	5.25	283.92	349.16
2018	1.023	1.023	0.847	0.939	0.901	1.000
	26.41	35.63	186.42	2.86	27.46	349.40
2019	1.012	1.012	0.928	0.978	0.964	1.000
	126.83	47.32	183.39	1.57	129.42	349.63
2020	0.999	0.999	1.023	1.018	1.028	1.000
	227.47	59.24	182.50	1.36	230.03	349.87

V_0+u (in degrees) at the beginning of years 1981-2020.


```
      PROGRAM TIDTAB
C------------------------------------- Franco Stravisi 1983,1986,1989
C TIDAL TABLES.
C Computes and prints one year of astronomical tide
C for TRIESTE.
C Harmonic method with 7 components (M2,S2,N2,K2,K1,O1,P1).
C References:
C Stravisi F. (1983): "The IT method for the harmonic tidal
C      prediction", Boll. Oceanol. Teor. Appl., 1/3, 193-204.
C Stravisi F. (1986): "A Fortran program for the harmonic
C      tidal prediction", CNR Ist. Talassografico Trieste,
C      RF 14.
C-- SUBROUTINES required:
C      TIMCOF    computes astronomical parameters;
C      HSPLON    computes moon and sun longitudes;
C      NLONG     computes longitude of moon's node;
C      INUXI     computes astronomical parameters;
C      NODFAC    computes the nodal factors;
C      VZEROU    computes Vo (1 Jan, 0h) and u (1 Jul);
C      ARCOTS    puts in COMMON the set of 7 harmonic
C                constants, nodal factors and phases;
C      HIGLOW    computes high/low waters;
C-- FUNCTIONS     required:
C    TIDE        computes the harmonic tide, 7 components;
C    D           computes the first time derivative of TIDE;
C    DD          computes the second time derivative of TIDE.
C-------------------------------------------------------------------
C    INPUT :  (from keyboard) year, code (1/2/3).
C             Version for IBM Proprinter.
C    OUTPUT : nodal factors and phases;
C             code = 1   ASCII file with hourly values
C             code = 2   prints monthly tables, hourly values
C             code = 3   prints monthly tab., high/low waters
--------------------------------------------------------------------
C    NOTE - changes for other localities:
C    1) define 7 harmonic constants H,G in sub. ARCOTS;
C    2) change tables headings in TIDTAB formats (1), 20, 31
C       and in ARCOTS format 1;
C    3) change names of months MESE and days GN
C       in TIDTAB DATA;
C    3) check local time headings in TIDTAB formats 26, 50.
C       Check control characters for printer:
C       IBM Proprinter II (80 col.) is here used as LPT1.
C-------------------------------------------------------------------
      DIMENSION AT(24,31,12),IAT(24),HP(1500),LP(1500),
     .IH(31,12,6),IM(31,12,6),IA(31,12,6),JS(31,12),
     .NG(12),NO(13)
      CHARACTER*9 MESE(12)
      CHARACTER*1 GN(7),DY(31,12)
      CHARACTER*30 FOUT
      DATA AT/8928*0.0/
      DATA IM/2232*100/
C-- NG : monthly no. of days
      DATA NG/31,28,31,30,31,30,31,31,30,31,30,31/
C-- MESE : names of months;  GN : days of the week.
      DATA MESE/'GENNAIO  ','FEBBRAIO ','MARZO    ',
```

```
     .           'APRILE    ','MAGGIO    ','GIUGNO    ',
     .           'LUGLIO    ','AGOSTO    ','SETTEMBRE',
     .           'OTTOBRE   ','NOVEMBRE  ','DICEMBRE  '/
      DATA GN/'L','M','M','G','V','S','D'/
C-- Input prediction year (NYEAR) and printing code (NCOD)
      WRITE(*,1)
    1 FORMAT(/
     .' University of Trieste, Italy'/
     .' Department of Theoretical Physics'/
     .' Program TIDTAB (Franco Stravisi 1983,1986,1989):'/
     .' harmonic tidal computations for TRIESTE'//
     .' Year ? ',\)
      READ(*,2) NYEAR
      NY=NYEAR-1900
      IF(NYEAR.GE.2000) NY=NYEAR-2000
    2 FORMAT(I5)
      WRITE(*,3)
    3 FORMAT(/' Code (1/2/3):'/
     .' 1 = fills a new Ascii file with hourly tide,'/
     .' 2 = prints  monthly tables with hourly tide,'/
     .' 3 = prints monthly tables with high/low waters;'/
     .' ?  '\)
      READ(*,2) NCOD
      IF(NCOD.EQ.1) GOTO 5000
      WRITE(*,4)
    4 FORMAT(
     .' !! LPT1 control codes are for IBM Proprinter.'/
     .' Print: default (1) or max qual. Letter II (2) ?  '\)
      READ(*,2) NST
C - Define harmonic parameters
 5000 CALL ARCOTS(NYEAR)
C    Modify number of days for leap years
      NG(2)=28
      IF(MOD(NYEAR,4).EQ.0) NG(2)=29
      IF(NCOD.EQ.1) GOTO 1000
      IF(NCOD.EQ.2) GOTO 500
C - Define days of the week DY (from 1 Jan 1976)
      NZ=4
      NDF=NYEAR-1976
      NZ=NZ+NDF*365+(NDF+3)/4-2
      DO 100 M=1,12
      DO 100 K=1,NG(M)
      NZ=NZ+1
  100 DY(K,M)=GN(MOD(NZ,7)+1)
  500 OPEN(1,FILE='LPT1')
C-- Set IBM Proprinter to Letter II, max quality
      IF(NST.EQ.2)
     .WRITE(1,5) CHAR(27),CHAR(73),CHAR(3),CHAR(27),CHAR(71)
    5 FORMAT(1X,5A1,\)
      IF(NCOD.EQ.3) GOTO 3000
C-- Compute hourly elevations AT (I=1,24 means  h = 0,23)
 1000 CONTINUE
      T=-1.
      DO 110 M=1,12
      DO 110 K=1,NG(M)
      DO 110 I=1,24
```



```
      T=T+1.
  110 AT(I,K,M)=TIDE(T)
      IF(NCOD.EQ.2) GOTO 2000
C-- (tide at December 31, h=24)
      ATF=TIDE(T+1.)
C-- Input NEW ASCII file name for output hourly tide
      WRITE(*,6)
    6 FORMAT(/' New Ascii file name for hourly tide ?  '\)
      READ(*,7) FOUT
    7 FORMAT(A30)
C-- Fill FOUT
      OPEN(2,FILE=FOUT,STATUS='NEW')
      DO 120 M=1,12
      DO 120 K=1,31
      IGG=K+M*100+NY*10000
      WRITE(2,10) IGG,(AT(I,K,M),I=1,12)
   10 FORMAT(I6,12F6.1)
      WRITE(2,11)     (AT(I,K,M),I=13,24)
   11 FORMAT(6X,12F6.1)
  120 CONTINUE
      IGG=101+(NY+1)*10000
      WRITE(2,12) IGG,ATF
   12 FORMAT(I6,F6.1)
      CLOSE(2)
      STOP
C-- Print monthly tables with hourly elevations
 2000 CONTINUE
      WRITE(*,15)
   15 FORMAT(/' Prints monthly tables with hourly tide.'/
     .' IBM Proprinter II: insert sheet with 2 LF.'/)
 2010 WRITE(*,16)
   16 FORMAT(' MONTH (1-12; 0=STOP) ?  '\)
      READ(*,2) M
      IF(M.EQ.0) GOTO 2100
      WRITE(1,20) CHAR(27),CHAR(69),MESE(M),NYEAR,
     .             CHAR(27),CHAR(70)
   20 FORMAT(14(/),14X,
     .'Università degli Studi di Trieste'/14X,
     .'Dipartimento  di  Fisica  Teorica'//1X,2A1//14X,
     .'TRIESTE - Marea astronomica /cm',9X,A9,I6,2A1)
      WRITE(1,21) CHAR(15)
   21 FORMAT(1X,A1\)
      WRITE(1,22) (I,I=0,23)
   22 FORMAT(/21X,'╔',96('═'),'╗'/21X,'║',24I4,'║',/
     .18X,'╔══╪',96('═'),'╡')
      DO 200 J=1,3
      JA=(J-1)*10+1
      JB=JA+9
      IF(J.EQ.3) JB=NG(M)
      DO 201 K=JA,JB
      DO 202 I=1,24
  202 IAT(I)=NINT(AT(I,K,M))
      WRITE(1,23) K,IAT
   23 FORMAT(18X,'|',I2,'|',24I4,'|')
  201 CONTINUE
      IF(J.LT.3) WRITE(1,24)
```

```
   24 FORMAT(18X,'|  |',96(' '),'|')
  200 CONTINUE
      IF(JB.EQ.31) GOTO 203
      DO 204 K=JB+1,31
  204 WRITE(1,24)
  203 CONTINUE
      WRITE(1,25)
   25 FORMAT(18X,'╚══╧',96('═'),'╝')
      WRITE(1,21) CHAR(18)
      WRITE(1,26)
   26 FORMAT(/14X,'Tempo Medio Europa Centrale (GMT + 1h).'/
     .14X,'Altezze riferite al livello medio del mare.')
      GOTO 2010
 2100 STOP
C - Compute high and low waters:
C   HP instants /h, from 0h 1 Jan, local time,
C   LP elevations /cm
 3000 CALL HIGLOW(NYEAR,HP,LP,JX)
      NO(1)=0
      DO 250 M=1,12
  250 NO(M+1)=NO(M)+NG(M)
C-- Translate HP into month, day, hour and minutes;
C-- define JS daily times IH/h, IM/min and elevations IA/cm
      DO 300 I=1,JX
      LH=HP(I)
      XH=LH
      XM=(HP(I)-XH)*60.
      IF(XM.LE.59.5) GOTO 301
      XM=0.
      LH=LH+1
  301 ID=LH/24+1
      MO=0
  302 MO=MO+1
      K=ID-NO(MO)
      IF(K.LE.0) GOTO 303
      GOTO 302
  303 K=K+NO(MO)
      M=MO-1
      K=K-NO(M)
      J=0
  304 J=J+1
      IF(IM(K,M,J).LT.100.AND.J.LT.6) GOTO 304
      JS(K,M)=J
      IH(K,M,J)=LH-(NO(M)+K-1)*24
      IM(K,M,J)=NINT(XM)
      IA(K,M,J)=LP(I)
  300 CONTINUE
C-- Print monthly tables with high/low waters
      WRITE(*,30)
   30 FORMAT(/
     .' Prints monthly tables with high/low waters.'/
     .' IBM Proprinter II: insert sheet with 2 LF.'/)
 3010 WRITE(*,16)
      READ(*,2) M
      IF(M.EQ.0) GOTO 3100
      WRITE(1,31) CHAR(27),CHAR(69),MESE(M),NYEAR,
```



```
     .             CHAR(27),CHAR(70)
   31 FORMAT(14(/),15X,'Università degli Studi di Trieste'/
     .15X,'Dipartimento  di  Fisica  Teorica'//1X,2A1
     .//15X,'TRIESTE - Alte e basse maree',8X,A9,I6,2A1//
     .19X,'╔',46('═'),'╗'/
     .19X,'║','   ora  cm',
     .    3('    ora  cm'),'║'/14X,'╔════╪',46('═'),'╣')
      DO 310 K=1,NG(M)
      JSF=JS(K,M)
      JB=MIN0(4,JSF)
      IF(JB.EQ.1) WRITE(1,41) K,DY(K,M),
     .             IH(K,M,1),IM(K,M,1),IA(K,M,1)
   41 FORMAT(14X,'|',I2,1X,A1,'|',I3.2,':',I2.2,I4,36X,'|')
      IF(JB.EQ.2) WRITE(1,42) K,DY(K,M),
     .            (IH(K,M,J),IM(K,M,J),IA(K,M,J),J=1,JB)
   42 FORMAT(14X,'|',I2,1X,A1,'|',I3.2,':',I2.2,I4,
     .                          I5.2,':',I2.2,I4,24X,'|')
      IF(JB.EQ.3) WRITE(1,43) K,DY(K,M),
     .            (IH(K,M,J),IM(K,M,J),IA(K,M,J),J=1,JB)
   43 FORMAT(14X,'|',I2,1X,A1,'|',I3.2,':',I2.2,I4,
     .                        2(I5.2,':',I2.2,I4),12X,'|')
      IF(JB.EQ.4) WRITE(1,44) K,DY(K,M),
     .            (IH(K,M,J),IM(K,M,J),IA(K,M,J),J=1,JB)
   44 FORMAT(14X,'|',I2,1X,A1,'|',I3.2,':',I2.2,I4,
     .                        3(I5.2,':',I2.2,I4),'|')
      IF(JSF.EQ.5) WRITE(1,45) IH(K,M,5),IM(K,M,5),IA(K,M,5)
   45 FORMAT(14X,'|',4X,'|',  I3.2,':',I2.2,I4 ,36X,'|')
      IF(JSF.EQ.6) WRITE(1,46)
     .              (IH(K,M,J),IM(K,M,J),IA(K,M,J),J=5,JSF)
   46 FORMAT(14X,'|',4X,'|',I3.2,':',I2.2,I4,I5.2,':',
     .                   I5.2,':',I2.2,I4,24X,'|')
  310 CONTINUE
      WRITE(1,50)
   50 FORMAT(14X,'╚════╧',46('═'),'╝'
     .//15X,'Tempo Medio Europa Centrale (GMT + 1h).'/
     .15X,'Altezze riferite al livello medio del mare.')
      GOTO 3010
 3100 STOP
      END


      FUNCTION TIDE(T)
C----------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes the astronomical tide by adding
C  7 harmonic components:
C             1    2    3    4    5    6    7
C            M2   S2   N2   K2   K1   O1   P1
C  S      angular speed in rad/h ;
C  FH     nodal factor times harmonic amplitude in cm ;
C  VUG    phase     Vo + u - g ;
C  FHS    FH times S  (to be used by D) ;
C  FHSS   FH times S x S  (to be used by DD) ;
C  T      local solar mean time in hours from 0 h, 1 Jan.
C---------------------------------------------------------
      COMMON /FTIDE/S(7),FH(7),VUG(7),FHS(7),FHSS(7)
```

```
      X=0.
      DO 10 K=1,7
   10 X=X+FH(K)*COS(S(K)*T+VUG(K))
      TIDE = X
      RETURN
      END
```

```
      FUNCTION D(T)
C------------------------ Franco Stravisi 1983,1986,1989 ---
C  First time derivative of TIDE .
C-----------------------------------------------------------
      COMMON /FTIDE/S(7),FH(7),VUG(7),FHS(7),FHSS(7)
      X=0.
      DO 10 K=1,7
   10 X=X-FHS(K)*SIN(S(K)*T+VUG(K))
      D = X
      RETURN
      END
```

```
      FUNCTION DD(T)
C------------------------ Franco Stravisi 1983,1986,1989 ---
C  Second time derivative of TIDE .
C-----------------------------------------------------------
      COMMON /FTIDE/S(7),FH(7),VUG(7),FHS(7),FHSS(7)
      X=0.
      DO 10 K=1,7
   10 X=X-FHSS(K)*COS(S(K)*T+VUG(K))
      DD=X
      RETURN
      END
```

```
      SUBROUTINE ARCOTS(NYEAR)
C------------------------ Franco Stravisi 1983,1986,1989 ---
C  Defines harmonic constants for Trieste (7 components):
C  S   angular speeds in rad/h;
C  MA  symbols of 7 constituent tides;
C  F   nodal factors;
C  H   hc amplitudes in cm;
C  VU  Vo + u  in rad;
C  G   hc phases in rad.
C-----------------------------------------------------------
      COMMON /FTIDE/S(7),FH(7),VUG(7),FHS(7),FHSS(7)
      DIMENSION F(7),H(7),VU(7),G(7)
      CHARACTER*2 MA(7)
      DATA MA/'M2','S2','N2','K2','K1','O1','P1'/
      DATA H/26.7,16.0,4.5,4.3,18.2,5.4,6.0/
      DATA G/4.843,4.993,4.798,4.993,1.241,1.066,1.241/
      S(1)=0.50586805
      S(2)=0.52359878
      S(3)=0.49636692
```



```
      S(4)=0.52503234
      S(5)=0.26251617
      S(6)=0.24335188
      S(7)=0.26108261
      Y=NYEAR
      YH=Y+0.5
      CALL NODFAC(YH,F)
      CALL VZEROU(Y,YH,VU,1)
      DO 10 L=1,7
      FH(L)=F(L)*H(L)
      FHS(L)=FH(L)*S(L)
      FHSS(L)=FHS(L)*S(L)
      VUG(L)=VU(L)-G(L)
   10 CONTINUE
      WRITE(*,1) NYEAR
      DO 20 I=1,7
   20 WRITE(*,2) MA(I),F(I),H(I),S(I),VU(I),G(I)
      RETURN
    1 FORMAT(//' (Subroutine ARCOTS)'//
     .' ASTRONOMIC TIDE          -        TRIESTE',I11
     .//1X,'h = f.H.cos(S.t+Vo+u-g)' //
     .10X,1Hf,7X,1HH,9X,1HS,9X,5HVo+u ,5X,1Hg/15X,'(cm)',
     .5X,'(rad/ora)',2X,2(3X,'(rad)')/)
    2 FORMAT(1X,A2,F10.3,F6.1,F14.8,1X,F9.5,F8.3)
      END
```

```
      SUBROUTINE HSPLON(Y,DL,IRD)
C------------------------ Franco Stravisi 1983,1986,1989 ---
C  Computes lunisolar longitudes as a function of time.
C  INPUT :  Y  time: year.(fraction of year)
C  OUTPUT:  DL(1)   mean longitude of sun, h       REAL*8
C           DL(2)   mean longitude of moon, s        "
C           DL(3)   longitude of lunar perigee, p    "
C  Parameter:    IRD  = 0  output in degrees
C                       1  output in rad
C-----------------------------------------------------------
      DOUBLE PRECISION T,T2,T3,DL(3)
      Y0=Y-1900.
      TR=FLOAT((INT(Y0)-1)/4)+0.5
C  Time in Julian centuries
      T=(3.65D2*Y0+TR)/3.6525D4
      T2=T**2
      T3=T**3
      DL(1)=2.79696678D2+3.6000768925D4*T+3.025D-4*T2
      DL(2)=2.70437422D2+4.81267892000D5*T+2.525D-3*T2
     .      +1.89D-6*T3
      DL(3)=3.34328019D2+4.069032206D3*T-1.0344D-2*T2
     .      -1.25D-5*T3
C  Reduce to (0,360)
      DO 10 I=1,3
   10 DL(I)=DMOD(DL(I),3.60D2)
      IF(IRD.EQ.0) RETURN
C  Degrees to rad
      DO 20 I=1,3
```

```
20 DL(I) DL(I)*3.14159265359D0/1.80D2
   RETURN
   END


   SUBROUTINE NLONG(Y,DN,IRD)
C----------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes longitude of moon's node.
C  INPUT :    Y   time:  year.(fraction of year)
C  OUTPUT:    DN  longitude of moon's node, N      REAL*8
C  Parameter: IRD  = 0  output in degrees
C                    1  output in rad
C-----------------------------------------------------------
   DOUBLE PRECISION T,T2,T3,DN
   Y0=Y-1900.
   TR=FLOAT((INT(Y0)-1)/4)+0.5
C  Time in Julian centuries
   T=(3.65D2*Y0+TR)/3.6525D4
   T2=T**2
   T3=T**3
   DN=2.59182533D2-1.934142397D3*T+2.106D-3*T2+2.22D-6*T3
 1 DN=DN+3.60D2
   IF(DN.LT.0.D0) GOTO 1
   IF(IRD.EQ.0) RETURN
   DN=DN*3.14159265359D0/1.80D2
   RETURN
   END


   SUBROUTINE INUXI(Y,DX,IRD)
C----------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes time dependent parameters.
C  INPUT :  Y  time:  year.(fraction of year)
C  OUTPUT:  DX(1)  obliquity of moon's orbit, I
C           DX(2)  right ascension of lunar intersection, nu
C           DX(3)  auxiliary term for K1, nu'
C           DX(4)  auxiliary term for K2, 2 nu"
C           DX(5)  long. in moon's orbit of lunar
C                  intersection, xi
C                  DX  are  DOUBLE PRECISION
C  Parameter:  IRD  = 0  output in degrees
C                   = 1  output in rad
C-----------------------------------------------------------
   DOUBLE PRECISION A(6),B(7),DX(5),C(6),DN,AR
   CALL TIMCOF(Y,A,B)
   CALL NLONG(Y,DN,1)
   C(1)=DN/2.D0
   C(2)=DSIN(C(1))/DCOS(C(1))
   C(3)=DATAN(A(3)*C(2))
   AR=A(1)-A(2)*DCOS(DN)
   DX(1)=DACOS(AR)
   C(4)=DSIN(2.D0*DX(1))
   C(5)=DSIN(DX(1))**2
   DX(2)=C(3)-DATAN(A(4)*C(2))
```



```
   C(6)=2.D0*DX(2)
   DX(3)=DATAN(C(4)*DSIN(DX(2))/(A(5)+C(4)*DCOS(DX(2))))
   DX(4)=DATAN(C(5)*DSIN(C(6))/(A(6)+C(5)*DCOS(C(6))))
   DX(5)=DN+DX(2)-2.D0*C(3)
   DO 10 I=1,5
   IF(DX(I).LT.0.D0) DX(I)=DX(I)+2.D0*3.14159265359D0
10 CONTINUE
   IF(IRD.EQ.1) RETURN
   DO 20 I=1,5
20 DX(I)=DX(I)*1.80D2/3.14159265359D0
   RETURN
   END


   SUBROUTINE VZEROU(Y,YH,VOU,IRD)
C----------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes the equilibrium phase at Greenwich at time t=0
C  for 7 harmonic tidal components.
C  INPUT :      Y   year.(fraction of year) at time t=0;
C               YH  year.(fraction of year) at the middle of
C                   the time interval, when u are evaluated.
C  OUTPUT:      VOU(1),..VOU(7)  for the harmonic tides:
C                                 1   2   3   4   5   6   7
C                                 M2  S2  N2  K2  K1  O1  P1
C                                   respectively.
C  Parameter: IRD  =  0     output in degrees
C                     1     output in rad
C-----------------------------------------------------------
   DOUBLE PRECISION DV(7),DL(3),DX(5),DR,AF,A4,A34
   DIMENSION VOU(7)
   AF=360.D0
   A4=90.D0
   A34=270.D0
   IF(IRD.EQ.0.) GOTO 1
   DR=3.14159265359D0/180.D0
   AF=AF*DR
   A4=A4*DR
   A34=A34*DR
 1 CALL HSPLON(Y,DL,IRD)
   CALL INUXI(YH,DX,IRD)
   DV(1)=2.D0*(DL(1)-DL(2)+DX(5)-DX(2))
   DV(3)=2.D0*(DL(1)+DX(5)-DX(2))-3.D0*DL(2)+DL(3)
   DV(4)=2.D0*DL(1)-DX(4)
   DV(5)=DL(1)+A4-DX(3)
   DV(6)=DL(1)-2.D0*DL(2)+A34+2.D0*DX(5)-DX(2)
   DV(7)=A34-DL(1)
   DO 10 I=1,7
   IF(I.EQ.2) GOTO 10
 2 DV(I)=DV(I)+AF
   IF(DV(I).LT.0.D0) GOTO 2
   VOU(I)=DMOD(DV(I),AF)
10 CONTINUE
   VOU(2)=0.
   RETURN
   END
```

```
      SUBROUTINE NODFAC(Y,F)
C---------------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes nodal factors for 7 harmonic tides.
C  INPUT :  Y  year.(fraction of year) ; usually defined
C              at middle of the time interval.
C  OUTPUT:  F(1),..F(7) nodal factors for the harmonic tides
C                       1   2   3   4   5   6   7
C                       M2  S2  N2  K2  K1  O1  P1
C                       respectively.
C----------------------------------------------------------------
      DIMENSION F(7)
      DOUBLE PRECISION A(6),B(7),DX(5),C(4)
      CALL INUXI(Y,DX,1)
      CALL TIMCOF(Y,A,B)
      C(1)=DSIN(DX(1))
      C(2)=C(1)**2
      C(3)=DSIN(2.D0*DX(1))
      C(4)=DCOS(DX(1)/2.D0)**2
      F(1)=B(1)*C(4)**2
      F(2)=1.
      F(3)=F(1)
      F(4)=DSQRT(B(3)*C(2)**2+B(6)*C(2)*DCOS(2.D0*DX(2))
     .          +B(7))
      F(5)=DSQRT(B(2)*C(3)**2+B(5)*C(3)*DCOS(DX(2))+B(7))
      F(6)=B(4)*C(1)*C(4)
      F(7)=1.
      RETURN
      END
```

```
      SUBROUTINE TIMCOF(Y,A,B)
C---------------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes auxiliary coefficients slowly variable with time
C  (see Stravisi 1983, Tab. Ic).
C  INPUT :         Y    time: year.(fraction of year)
C  OUTPUT:         A(1),...A(6)    DOUBLE PRECISION
C                  B(1),...B(7)    DOUBLE PRECISION
C----------------------------------------------------------------
      DOUBLE PRECISION A(6),B(7),C(16),P
C  S'
      C(1)=4.5987564D-1
C  e
      C(2)=5.490056D-2
C  i/deg
      C(3)=5.14537628D0
      C(4)=Y-1900.
C  e1
      C(5)=1.675104D-2-4.180D-7*C(4)-1.26D-11*C(4)**2
C  ecliptic obl./deg
      C(6)=2.3452294D1-1.30111D-4*C(4)
C  pi
      P=3.14159265359D0
      P=1.80D2/P
      C(7)=P
      C(8)=DSIN(C(3)/P)
      C(9)=DSIN(C(6)/P)
```



```
      C(10)=(C(6)-C(3))/2.D0
      C(11)=(C(6)+C(3))/2.D0
      C(12)=DSIN(C(6)*2.D0/P)
      C(13)=DCOS(C(6)*0.5D0/P)
      C(14)=DCOS(C(3)*0.5D0/P)
      C(15)=1.D0-1.5D0*C(8)**2
      A(1)=DCOS(C(3)/P)*DCOS(C(6)/P)
      A(2)=C(8)*C(9)
      A(3)=DCOS(C(10)/P)/DCOS(C(11)/P)
      A(4)=DSIN(C(10)/P)/DSIN(C(11)/P)
      C(16)=C(1)*(1.D0+1.5D0*C(5)**2)/(1.D0+1.5D0*C(2)**2)
      A(5)=C(16)*C(12)
      A(6)=C(16)*C(9)**2
      B(1)=(C(13)*C(14))**(-4)
      B(2)=(A(5)+C(15)*C(12))**(-2)
      B(3)=(A(6)+C(15)*C(9)**2)**(-2)
      B(4)=C(13)**(-2)*C(14)**(-4)/C(9)
      B(5)=2.*A(5)*B(2)
      B(6)=2.*A(6)*B(3)
      B(7)=B(2)*A(5)**2
      RETURN
      END
```

```
      SUBROUTINE HIGLOW(NY,HP,LP,NHL)
C---------------------------- Franco Stravisi 1983,1986,1989 ---
C  Computes high/low water instants and elevations by means
C  of the Newton-Fourier iteration.
C  INPUT :   NY   calendar year.
C  OUTPUT:   HP   vector filled with the high/low water
C                 instants in hours, accuracy EP < 0.5 min;
C            LP   integer vector filled with the high/low
C                 water elevations in cm, rounded to 1 cm;
C            NHL  total no. of high/low waters in the year.
C  This subroutine requires:
C  FUNCTION TIDE(T)  computes the astronomical tide in cm,
C  FUNCTION D(T)     computes the first time derivative,
C  FUNCTION DD(T)    computes the second time derivative
C                    of the predicted astronomical tide
C                    as a function of time T in hours.
C----------------------------------------------------------------
      DIMENSION HP(1500),LP(1500)
      ND=365
C  Leap years
      IF((NY/4)*4.EQ.NY) ND=366
C  Time resolution in hours between high/low waters
      S=1.
C  Time accuracy epsilon < 0.5 min
      EP=0.0083
C  Sigma
      SI=S/5.
      HY=ND*24
      LX=HY/S+2.
      J=1
      HP(J)=0.
```

```
      D2=D(0.)
      DO 100 I=1,LX
      L=J-1
      TA=FLOAT(I-1)*S
      D1=D2
      D2=D(TA+S)
      DD1=D1*D2
      IF(DD1) 110,110,100
  130 TAU=TAU+SI
      T=TA+TAU
      IF(TAU-S) 120,120,100
  110 T=TA
  120 TO=T
      DD0=DD(TO)
      IF(DD0.EQ.0.) DD0=DD0+1.E-10
C  Newton-Fourier formula
      T=TO-D(TO)/DD0
      AB=ABS(T-TO)
      IF(AB.GT.SI) T=TO+SI
      IF(AB.GE.EP) GOTO 120
      IF((T+EP).LE.TA) GOTO 130
      HP(J)=T
      IF(L.EQ.0) GOTO 150
      IF(ABS(T-HP(L)).LE.EP) GOTO 100
  150 LP(J)=NINT(TIDE(HP(J)))
      NHL=J
      J=J+1
  100 TAU=0.
      RETURN
      END
```

```
10 REM - PROGRAM "PLOMAR": MONTHLY PLOT OF ASTRONOMICAL TIDE
20 REM - INPUT DATA: HOURLY TIDE FROM ASCII FILE CREATED BY
30 REM - FORTRAN PROGRAM "TIDTAB"  (FRANCO STRAVISI 1987).
40 DIM L(745),C(6),M$(12),FL(31,12),AT(24,31,12),NG(12)
50 NG(1)=31:NG(2)=28:NG(3)=31:NG(4) =30:NG(5) =31:NG(6) =30
60 NG(7)=31:NG(8)=31:NG(9)=30:NG(10)=31:NG(11)=30:NG(12)=31
70 FOR M= 1 TO 12: FOR I=1 TO 31: FL(I,M)=0: NEXT I: NEXT M
80 M$(1)="GENNAIO  ": M$(7)="LUGLIO    "
90 M$(2)="FEBBRAIO ": M$(8)="AGOSTO    "
100 M$(3)="MARZO    ": M$(9)="SETTEMBRE"
110 M$(4)="APRILE   ": M$(10)="OTTOBRE  "
120 M$(5)="MAGGIO   ": M$(11)="NOVEMBRE "
130 M$(6)="GIUGNO   ": M$(12)="DICEMBRE "
140 PRINT " MONTHLY PLOT OF THE ASTRONOMICAL TIDE FOR
TRIESTE"
150 PRINT " TO CHANGE MONTH, RESTART BY TYPING <GOTO 290>"
160 INPUT " YEAR "; IA
170 INPUT " ASCII FILE WITH HOURLY TIDE "; NF$
180 INPUT " ASCII FILE WITH LUNAR PHASES"; NFM$
190 IF INT(IA/4)*4=IA THEN NG(2)=29
200 OPEN "I", #2, NFM$
210 INPUT #2, IAN,NPH
220 IF IAN<>IA THEN PRINT "ERROR IN ";NFM$
```



```
230 FOR I=1 TO NPH: INPUT #2, MM,DD,HH,IPH:FL(DD,MM)=IPH
   :NEXT I
240 OPEN "I", #1, NF$
250 FOR MM=1 TO 12:FOR K=1 TO 31: INPUT #1,GG :FOR I=1 TO 24
260 INPUT #1, AT(I,K,MM)
270 NEXT I: NEXT K: NEXT MM
280 INPUT #1, GG,ATF
290 INPUT " MONTH "; M
300 NM=NG(M)*24+1: J=0
310 FOR K=1 TO NG(M): FOR I=1 TO 24: J=J+1
320 L(J)=AT(I,K,M): NEXT I: NEXT K
330 IF M<>12 THEN L(NM)=AT(1,1,M+1)
340 IF M=12 THEN L(NM)=ATF
350 FOR I=1 TO 6: C(I)=680-120*(I-1): NEXT I
360 DEF FNX(X)=4*X+63
370 DEF FNY(Y)=Y/2+2
380 CLS: SCREEN 3: KEY OFF
390 WINDOW (1,1)-(640,400)
400 HA=0: HB=144: XA=FNX(HA): XB=FNX(HB)
410 LA=0: LB=760: YA=FNY(LA): YB=FNY(LB)
420 LINE(XA,YA)-(XB,YA): LINE-(XB,YB): LINE-(XA,YB)
   :LINE-(XA,YA)
430 LOCATE 3,14:PRINT"1";:LOCATE 3,26:PRINT"2";:LOCATE 3,38
   :PRINT"3";
440 LOCATE 3,50: PRINT"4";: LOCATE 3,62: PRINT"5";
   :LOCATE 3,74: PRINT"6";
450 LOCATE 6,14: PRINT"7";:LOCATE 6,26:PRINT"8";:LOCATE 6,38
   :PRINT"9";
460 LOCATE 6,49:PRINT"10";:LOCATE 6,61:PRINT"11";
   :LOCATE 6,73 :PRINT"12";
470 LOCATE 10,13:PRINT"13";:LOCATE 10,25:PRINT"14";
   :LOCATE 10,37 :PRINT"15";
480 LOCATE 10,49:PRINT"16";:LOCATE 10,61:PRINT"17";
   :LOCATE 10,73 :PRINT"18";
490 LOCATE 14,13:PRINT"19";:LOCATE 14,25:PRINT"20";
   :LOCATE 14,37 :PRINT"21";
500 LOCATE 14,49:PRINT"22";:LOCATE 14,61:PRINT"23";
   :LOCATE 14,73 :PRINT"24";
510 LOCATE 18,13:PRINT"25";:LOCATE 18,25:PRINT"26";
   :LOCATE 18,37 :PRINT"27";
520 LOCATE 18,49:PRINT"28";
530 IF NG(M)=28 THEN GOTO 590
540 LOCATE 18,61:PRINT"29";
550 IF NG(M)=29 THEN GOTO 590
560 LOCATE 18,73:PRINT"30";
570 IF NG(M)=30 THEN GOTO 590
580 LOCATE 22,13:PRINT"31";
590 FOR D=1 TO NG(M)
600 IF FL(D,M)<>0 THEN GOSUB 1050
610 NEXT D
620 FOR I=1 TO 5: YI=FNY(C(I)): LINE(XA,YI)-(XB,YI): NEXT I
630 LINE(XA,FNY(C(6)))-(FNX(24),FNY(C(6)))
640 LINE(FNX(31.5),YA)-(FNX(31.5),FNY(120))
650 LINE(FNX(31.5)+1,YA)-(FNX(31.5)+1,FNY(120))
660 LINE(FNX(31.5),FNY(120))-(XB,FNY(120))
670 LINE(FNX(31.5),FNY(120)-1)-(XB,FNY(120)-1)
```

```
680 LINE(FNX(24),YB)-(FNX(24),YA)
690 LINE(FNX(31.5),YA+1)-(XB,YA+1)
   :LINE(XB-1,YA)-(XB-1,FNY(120)-1)
700 FOR I=1 TO 4: XI=FNX((I+1)*24)
   :LINE(XI,YB)-(XI,FNY(120)): NEXT I
710 JM=6: IF NG(M)<31 THEN JM=5
720 H2=1 : FOR J=1 TO 6: H1=H2: HL=144
730 IF J=5 AND NG(M)<30 THEN HL=(NG(M)-24)*24
740 IF J=6 THEN HL=24
750 H2=H1+HL
760 DL=0
770 IF J>JM THEN GOTO 820
780 X=XA: Y=FNY(L(H1)+C(J)+DL): PSET(X,Y) : CIRCLE(X,Y),1
790 FOR H=H1+1 TO H2
800 X=X+4:Y=FNY(L(H)+C(J)+DL):LINE-(X,Y):CIRCLE(X,Y),1
810 NEXT H
820 HD=H2-6: IF J=6 THEN HD=18
830 FOR H=6 TO HD STEP 6: YU=FNY(C(J)+4): YD=FNY(C(J)-4)
   :FH=FNX(H)
840 LINE(FH,YU)-(FH,YD): NEXT H
850 HD=144: IF J=6 THEN HD=24
860 FOR H=0 TO HD STEP 24: XS=FNX(H-.5): XD=FNX(H+.5)
870 IF H=0 THEN XS=FNX(0)
880 IF H=HD AND J<6 THEN XD=FNX(H)
890 FOR IS=-40 TO 40 STEP 40: YS=FNY(IS+C(J))
   :LINE(XS,YS)-(XD,YS): NEXT IS
900 NEXT H
910 NEXT J
920 X=163: Y=22
930 LINE(X+1,Y)-(X+4,Y)
940 LINE(X+10,Y-3)-(X+10,Y+3): LINE-(X+6,Y-1)
   :LINE-(X+11,Y-1)
950 Y=62
960 LINE(X+10,Y-3)-(X+10,Y+3): LINE-(X+6,Y-1)
   :LINE-(X+11,Y-1)
970 FOR Y=22 TO 62 STEP 20: CIRCLE(X+17,Y),3,,,,1.5: NEXT Y
980 LOCATE 23,26
990 PRINT "Università di Trieste, Dipartimento di Fisica
Teorica";
1000 LOCATE 24,26: PRINT "TRIESTE  - Marea astronomica
/cm";
1010 LOCATE 24,64: PRINT USING "\        \"; M$(M);: PRINT
USING"######"; IA;
1020 LOCATE 1,1: FOR I=1 TO 15: LOCATE,,,50+I,0: NEXT I
1030 GOTO 1140
1040 REM - PLOT LUNAR PHASES
1050 J=INT((D-1)/6)+1: H=(D-1)*24-(J-1)*144+18: X=FNX(H)
1060 HL=(D-1)*24+18: YL=60
1070 IF D > 6 THEN YL=(L(HL+1)+L(HL-143)+120)/2
1080 Y=FNY(C(J)+YL): CIRCLE(X,Y),7
1090 IF FL(D,M)=1 THEN PAINT(X,Y)
1100 IF FL(D,M)<>3 THEN LINE(X,Y-5)-(X,Y+5)
1110 IF FL(D,M)=2 THEN PAINT(X-2,Y)
1120 IF FL(D,M)=4 THEN PAINT(X+2,Y)
1130 RETURN
1140 END
```